Appendix I- Columns of the Phase I Spreadsheet

Link to the Spreadsheet

The Phase I spreadsheet has 56 columns and is divided into eight worksheets. Each worksheet is seven years and four months long. A year consists of 12 months, and each month consists of 30 days. Actual years are 4 or 5 days longer, so the time to reach a goal will be slightly less than shown in the spreadsheet.

Each worksheet can have unique parameters and need not be used for the entire seven years and four months. To continue processing on the next worksheet, the last segment of the last day of the month before the final month plus the entire final month is copied BY VALUE to the start of the next worksheet.

The first four segments simulate a hypothetical 20-year Phase I. The fifth segment is used to freeze land purchases, clear the 99.16% inventory, and fill out the entire 20 years as input to the Phase II spreadsheet. The sixth segment is unused. Segments can be reused for an arbitrarily long Phase I if one is willing to lose intermediate information. A separate spreadsheet simulates a two-century Phase I from natural demand alone by overlaying intermediate information. The first three segments in that spreadsheet start, and the last three segments finish the 236 years.

For comparison purposes, the Phase I termination metric is defined as 190,000 million $250,000 average parcels in 2022 dollars and a minimum of $80 trillion in property value. In an implementation, Phase I cannot terminate until no more property owners are willing and able to sell into the commons trust.

The period is one day long, which more closely matches actual implementations than a monthly period. Market makers expect to turn over inventory in a day. They can use the one-day periods for optimal inventory planning. Daily compounding of dividends accurately reflects an implementation.

Each day is divided into five segments, A - E. Five segments are sufficient to handle the money flow during the period without ambiguity. Segments that divide a day and the segments that represent each worksheet in the spreadsheet should not be confused.

The diagram below shows the first 18 columns of the initial worksheet. The first two days and the last two days are shown. The parameters on this worksheet were

set very high for the first year of operation. Unused results from the last two days of year seven and month four show that almost every property in the world has been purchased into the commons trust. It is not probable that the retail demand as a percent of property value cannot be maintained for seven years.

Period	Y	M	D	Segment	MM Demand (Mil $)	Percent Peg	Dividend Percent	Retail Supply Shock(+) Demand Shock(-)	MM Supply Shock(+) Demand Shock(-)	Buy and Destroy Elsies demand shock (-)	Rescue Mode	Ongoing Retail Trade	Ram and Jam Contracts Signed	Land Fund (Mil $)	Land Fund Loan	Property Value (Mil $)	Total Purchases (number of properties)
0		0	0	E	$6.00	99.00%	7.00%	0	0		$0	$0		$1.00		$0	
1	2025	1	1	A	$6.00	99.00%	7.00%	0	0	0.00	$0	$4	$6.00	$0.70	0	$6	24
1	2025	1	1	B	$4.36	99.00%	7.00%	0	0	0.00	$0	$4	$0.00	$0.70	0	$6	24
1	2025	1	1	C	$4.36	99.00%	7.00%	0	0	0.00	$0	$4	$0.00	$0.70	0	$6	24
1	2025	1	1	D	$4.36	99.00%	7.00%	0	0	0.00	$0	$4	$0.00	$0.70	0	$6	24
1	2025	1	1	E	$4.36	99.00%	7.00%	0		0.00	$0	$4	$0.00	$0.70	0	$6	24
2	2025	1	2	A	$4.36	99.00%	7.00%	0	0	0.00	$0	$7	$4.36	$0.41	0	$10	41
2	2025	1	2	B	$3.24	99.00%	7.00%	0	0	0.00	$0	$7	$0.00	$0.41	0	$10	41
2	2025	1	2	C	$3.24	99.00%	7.00%	0	0	0.00	$0	$7	$0.00	$0.41	0	$10	41
2	2025	1	2	D	$3.24	99.00%	7.00%	0	0	0.00	$0	$7	$0.00	$0.41	0	$10	41
2	2025	1	2	E	$3.24	99.00%	7.00%	0	0	0.00	$0	$7	$0.00	$0.41	0	$10	41
2639	2032	4	29	A	$1,072,194.49	99.00%	7.00%	0	0	0.00	$0	$207,186,752	$1,072,194.49	$3,357.17	0	$304,686,399	1,040,525,082
2639	2032	4	29	B	$867,473.66	99.00%	7.00%	0	0	0.00	$0	$207,186,752	$0.00	$3,357.17	0	$304,686,399	1,040,525,082
2639	2032	4	29	C	$1,398,820.01	99.00%	7.00%	0	0	0.00	$0	$207,186,752	$0.00	$3,357.17	0	$304,686,399	1,040,525,082
2639	2032	4	29	D	$1,398,820.01	99.00%	7.00%	0	0	0.00	$0	$207,186,752	$0.00	$3,357.17	0	$304,686,399	1,040,525,082
2639	2032	4	29	E	$1,402,477.01	99.00%	7.00%	-30469	0	0.00	$0	$207,186,752	$0.00	$3,357.17	0	$304,686,399	1,040,525,082
2640	2032	4	30	A	$1,432,945.65	99.00%	7.00%	0	0	0.00	$0	$208,039,789	$1,220,612.78	$219,171.33	0	$305,940,866	1,044,741,495
2640	2032	4	30	B	$966,939.06	99.00%	7.00%	0	0	0.00	$0	$208,039,789	$0.00	$219,171.33	0	$305,940,866	1,044,741,495
2640	2032	4	30	C	$1,570,488.89	99.00%	7.00%	0	0	0.00	$0	$208,039,789	$0.00	$219,171.33	0	$305,940,866	1,044,741,495
2640	2032	4	30	D	$1,570,488.89	99.00%	7.00%	0	0	0.00	$0	$208,039,789	$0.00	$219,171.33	0	$305,940,866	1,044,741,495
2640	2032	4	30	E	$1,574,199.49	99.00%	7.00%	-30594	0	0.00	$0	$208,039,789	$0.00	$219,171.33	0	$305,940,866	1,044,741,495

A start year is set at the top of the worksheet. In this case, 2025. Column 2, with a "Y" header, shows the year. The next column shows the month in the year (1-12), and the following column shows the day in the month (1-30). The worksheet ends on the 30th day of April, 2032. Remember that twelve 30-day months only approximate a year, and February 30th might lead to confusion.

The five segments per day, A-E, are shown in the next column.

The first column, Period, is the day of Phase I. The spreadsheet begins with the last segment of period 0 to initialize the column. Formulas for the column begin in segment A of period 1. In subsequent displays, columns that are not germane to the discussion are hidden.

Market maker demand (column 6), initialized to $6 million, is entered by hand and can be changed for different simulations. All U.S. dollar and Elsie columns are in millions. "The Number of Properties Purchased" in column 18 is inflation-independent and reliable for determining the end of Phase I, regardless of duration.

Advanced rent mode, which only levied 5% of the purchase price for the advance rent fund, is no longer seen as a viable option in Phase I. Instead, any property, regardless of land share, qualifies for the new direct purchase mode, where 55% of the purchase price is levied. 5% goes to the advance rent fund, and 50% is disbursed with an "immediate-delayed" disposition. Because this is equivalent to

an auction following a sale, auction mode (formerly sales mode) is assumed throughout the simulation.

Columns will be described in the order in which they appear in the spreadsheet. The formula for the column will be boxed, highlighted, and described in detail, including background theory, if needed.

The spreadsheet is double-zero-balanced. The U.S. dollars input into the system are precisely equal to the U.S. dollars in the various paid and current accounts. The Elsies created from purchases are identical to the Elsies in the paid and current accounts, plus the Elsies destroyed. These balanced columns appear in green near the final columns of the spreadsheet. Here is an example from day 800 (March 20th, 2027) of the 20-year Phase I simulation, with most columns hidden. The green columns are farther apart than shown.

Period	Y	M	D	Segment	MM Demand (Mil $)	Ram and Jam Contracts Signed	Land Fund (Mil $)	Total Purchases (number of properties)	EDSF	Total Accounts	Elsies Created	Elsies Destroyed	Total Elsies	Total Dollars Input (Mil $)	ABC (Mil $)	Total Account Dollars (Mil $)
800	2027	3	20	A	$31.46	$31.46	$6.89	61,432	1,898.80	14,203	15,822.85	1,619.91	14,203	$19,143	$422.38	$19,143
800	2027	3	20	B	$22.78	$0.00	$6.89	61,432	1,898.80	14,203	15,822.85	1,619.91	14,203	$19,160	$422.38	$19,160
800	2027	3	20	C	$33.00	$0.00	$6.89	61,432	1,898.80	14,203	15,822.85	1,619.91	14,203	$19,160	$422.38	$19,160
800	2027	3	20	D	$33.00	$0.00	$6.89	61,432	1,898.80	14,203	15,822.85	1,619.91	14,203	$19,160	$422.38	$19,160
800	2027	3	20	E	$33.11	$0.00	$6.89	61,432	1,903.07	14,203	15,822.85	1,619.91	14,203	$19,160	$423.40	$19,160
801	2027	3	21	A	$33.11	$33.11	$6.32	61,559	1,903.07	14,232	15,855.96	1,624.45	14,232	$19,184	$423.40	$19,184
801	2027	3	21	B	$23.66	$0.00	$6.32	61,597	1,905.92	14,242	15,865.96	1,624.45	14,242	$19,211	$423.40	$19,211

Notice that there are purple-headed and red-headed columns on the spreadsheet. Purple-headed columns are dollar account columns (their sum adds to the Total Account Dollars, the final column shown above). Red-headed columns are Elsie-account columns. Their sum is added to the Total Accounts column (the first green column above). Only one purple-headed (ABC) and red-headed column (EDSF) are shown in the table above.

Modifiable Parameters

				Periods per month		30			
				Starting Average Property Price		0.25			
				General Inflation rate		2%	Commons land Appreciation		4.0%
				Owners allowing rent to fall times percent of treblers using Elsies		50%			
				Loan Criterion		4.0	Elsie Destroy Game		0.00
				Retail Demand/Savers		68.00%			
				Sequestered Elsie Bidders per Property		0.00	Probability of Elsie bid win		0.00%
				MM Bottom		99.05%		MM Top	99.15%
				Start Year		2025			

These parameters at the top of each worksheet significantly affect the outcome. Values shown are for the first year of the 20-year Phase I. “Starting Average Property Price” is always set to $0.25, representing $250,000. Any other value will change the endpoint for Phase I.

The general inflation rate is 2%, and land or property appreciates at 4% or 2% over inflation. Due to the lack of property taxes, growth is expected to be 5% in the second year, but not in the starting segment. A value of 6% represents the formation of megacities on commons trust land, and 8% represents super-megacities. It is unrealistic for growth rates on average to exceed 8%. Not all commons trust property will become megacities. Conservative simulations should be run at 4-5.5%.

Owners allowing rent to fall times the percentage of treblers using Elsies is set at 50%**.** If the goal is to pay the lowest rent, the most efficient rental strategy is to allow one's rent to fall until trebled and then match the trebler (or surrender the property). However, for those at the very top, high rents bring aristocracy and bragging rights (in land-based capitalism, rents are the new Rolex). For those who live paycheck to paycheck, lack of liquidity prevents matching the trebler. Some would rather pay monthly rent and avoid the hassle of being trebled every year. Not everyone will use an efficient rental strategy.

While treblers tend to be more sophisticated than property bidders at auction and more invested in the AFFEERCE business plan, there is no assurance that all of them will use Elsies to take advantage of the 0.85% arbitrage discount.

If everybody used the efficient rental strategy, and all trebles used Elsies, the value of this parameter should be set to 100%. If 75% of the property owners used the efficient rental strategy and 66.67% of treblers used Elsies, it should be set to (75% of 66.67% =) 50%. This is the value used for natural demand, but in the second year of a very active Phase I, the parameter has increased to 60%. During deep discounting (a fall below 99% of peg), a value of 100% is automatically used.

The parameter "Retail demand/savers" (retail_demand or Q8) is set initially to 68% of the total property value in this simulation. Retail demand includes any member of the public holding Elsies, whether for saving, trebling, bidding, commerce, or the dividend. For natural demand simulations, there is no retail demand.

Numbers above 75% can be considered insatiable demand, although the discrete nature of the spreadsheet and the 30-day delay to closing will show this demand still has a duration. In the first table displayed above, this parameter is set to 68%, and by April of 2032, $309.5 trillion of properties were in the commons trust

(almost the total world's property). We only used one year at 68% for the 20-year Phase I simulation.

A marker of higher-than-possible retail demand is segment A's negative market maker inventory. While the market maker's stock often goes negative in segment B, if it cannot recover by segment A of the next period, there is demand for more Elsies than are minted for a property or that must be sequestered. In that case, insatiable demand and logistics are the only constraints on the duration of Phase I. The logistical limit is a function of the property in the commons trust. At the very start of Phase I, "Retail demand/savers" can be as high as 90% for a few months before logistics becomes the only constraint. Logistics will be the only constraint after 150 years of natural demand if the parameter rises to 45% on a continuation worksheet. The more prolonged Phase I has muddled along with natural demand, the easier it is to start an Elsie buyer's panic.

When estimating the value for "Retail demand/savers," consider the number of people holding Elsies for retirement, the number of people keeping Elsies for the dividend, and the extent to which Elsie is used in commerce. This also includes other market makers not accounted for in the simulation. Legacy governments would be wise to purchase Elsies to protect their citizens from Phase II hyperdeflation.

The largest holders of Elsies are renters protecting themselves from a treble, professional auction bidders, with possibly the largest of all being professional treblers. Bidding at auction and trebling are free services. Matched Treblers usually have their money freed in 3-6 days. The funds are sequestered until possession changes in the event of treble success.

Professional bidders might enter a low bid at multiple auctions, hoping for a win. With the winning bid expected to be near 50% of property value, the Elsies held by a professional bidder might be three times the average property value for six concurrent actions. Treblers need about 150% of the property value (133% of the structure plus treble rent). Professional treblers would hold three times the average property value for two concurrent trebles. If 0.1% of the world's population were professional bidders or treblers, they would demand enough Elsies (8 million x 3 x $250,000 = 6 trillion) to terminate Phase I as quickly as logistically possible and bring about the hyperdeflation of Phase II in under a year.

There is every reason to believe those who do not possess Elsies in Phase II will suffer financial ruin. Even if this is an exaggeration, no portfolio should be without Elsies as the world enters Phase II. Proper communication of this message will significantly increase the holding of Elsies and bring Phase II much closer simultaneously, creating a "virtuous" cycle and even triggering panic buying of Elsies.

Because communication of these ideas is the crux of the value for the retail/savers parameter, consider using numbers from 0% to 63% as a measure of CEO skill. However, a famous CEO who hypes it to the world during the first year can set this parameter for one year only, at 68% to 70%.

This simulation sets the sequestered auction bids per property at 0 and then 10. Auction bids are sequestered for less than 24 hours (including the duration of segment C, where Elsies are qualified for the dividend). The only cost of an auction bid or treble is the lost dividends (usually not too high).

The probability of an Elsie bid winning a property auction is set for year two at 0% and rises in the third and fourth segments. Most auction bids are expected to be in U.S. dollars for most of a 20-year Phase I.

Loan Criterion is deprecated and should be kept at 4.0. The "Elsie Destroy Game" parameter 0.00 will be discussed when covering the "Buy and Destroy Elsies" column. For natural demand, it should always be 0.00.

MM Demand

=MIN (U259 – T259, U259)

The driving demand of the entire simulation is just the desired Elsie inventory(**U259**) minus the current Elsie inventory(**T259**), as described above. However, if the current inventory goes negative, the desired inventory increases to account for that. Demand that accounts for the difference will be twice the actual demand, so if T259 is negative, then the demand is just U259.

Percent of Peg

=IF (E260="A", MIN (99%, IF (F260>U259/1000, G259+0.02%, G259)), G259)

This evaluates to 99%, the default percent of peg where the Elsie trades. However, suppose the value in the previous period is lower(G259). In that case, it will use that value unless the current demand(**F260**) is positive and greater than the desired Elsie inventory of the market maker(**U259**) divided by 1000. If that is the case, it will add 0.02% to the previous value.

In the event of a material retail dump, demand will turn negative. If the red (negative values) extends for more than one month, the percent of the peg must be manually adjusted in segment A in the first month, which is all red. The goal is to find the highest percentage of peg, where the demand returns to black the following month. Once this is done, the peg percentage returns to 99% as fast as the market allows.

The lowest setting seen in tests is 93.2%. Recovery time can range from 2 months to 20 years. In the implementation, recovery times can vary from 2 seconds to 20 years. In all simulations, Phase I completes faster with a purchase and dump than if the purchase was never made. When retail is a fixed percentage of property value, Phase I can finish more quickly with the purchase and dump than the purchase alone. That is because retail is a percentage of property value that does not drop with the dump. Retail Elsies are restored at once, according to the formula, and purchase property for a second time.

=MIN (32%, 7% + (99% - G260) * 10)

Dividend Percent

During deep discounting, the dividend distribution increases at the EDSF's expense, up to a maximum dividend distribution of 32% of the rent. The formula is the base 7% plus 99% - the current percent of peg (**G260**) times 10.

Retail Demand and Supply Shocks

=IF (AND (E285="E", T284>0), IF (AP285>9%, MIN (-1, -1 * Q284/10000), IF (AP285>8%, MIN (-0.5, -0.5 * Q284/10000), IF (AP285>7%, MIN (-0.25, -0.25 * Q284/10000), IF (AP285>6%, MIN (-0.1, -0.1 * Q284/10000), 0)))), 0)

The formula has very little to do with retail demand and supply shocks. Instead, it creates a spike in demand whenever the dividend return exceeds specific yields. The annualized dividend must be at least 6% before this demand kicks in. The 6% should be converted to a parameter if this demand is considered material. Without strong retail demand, a high dividend creates retail demand. The formula enters a demand shock to be handled by the market maker. 6% seems reasonable based on the yield curve at the time of this writing. This demand for yield is considered natural demand because a certificate of deposit (in U.S. dollars) can be profitably issued to buyers once the dividend passes all points on the yield curve.

Changing the minimum dividend before natural demand kicks in from 6% to 7% increases the duration of Phase I from 228 years to 268 years for 5% growth and from 293 to 340 years for 4% growth. The formula can be deleted to test natural demand without dividend processing.

The formula is overwritten to set retail demand and supply shocks manually. Demand shocks are entered as negative numbers, and supply shocks are positive. In segment E, entering -500 means a demand for 500 Elsies from the market maker at 99.15% of the peg to be deposited in the retail Elsies column. Setting a value of 420 means a sale of 420 Elsies from retail to the market maker, purchased at 99.05%.

A demand shock manually entered in segment A, such as -2000, means a demand from the ABC for 2,000 Elsies produced during ram and jam. If the market maker inventory is insufficient, requests should be made in segment A. This reflects the implementation, where the market makers will handle rents and minor purchases of Elsies. Most will not have sufficient inventory for large purchases. Large purchasers are incentivized to buy directly from the ABC and win Elsies at 99.05%. The buyer must be prepared to wait for ram and jam and for the market makers to replenish their inventory.

In the formula, (**AP285**) refers to the annualized dividend. (**Q284**) is the current property value in the commons trust, and (**T284**) is the market maker Elsie inventory. In the simulation, inventory must be positive to satisfy dividend demand.

Market Maker Supply and Demand Shocks

0

This field can control the desired level of market maker inventory. It is not so much a supply shock as a new lower limit on inventory restocking. A value of +100 will decrease the market maker's desired inventory by 100. A value of -800 will increase the market maker's desired inventory column by 800. This is a demand shock that will trigger increased ram and jam. Values in this column are experimental and should be kept at 0 for most simulations.

Buy and Destroy Elsies

=IF (E260="E", -T7, 0)

The formula takes the “Destroy Elsies” parameter(**T7**) into a negative demand shock. It is demand on the market maker inventory. However, all Elsies purchased are destroyed. The formula can be overwritten in segment E with any negative demand shock. The purchase and destruction of Elsies, particularly early in Phase I, is the most significant possible accelerator of Phase I. It is a great sacrifice to the donor unless it can be packaged in a game with desirable prizes.

Rescue Mode

=L2561

The value manually entered in this rescue mode slot will repeat every ten days. Thus, segment B of day 3/10 will hold the value (**L2561**) found in the box above.

Users of the spreadsheet are encouraged to experiment with formulas and manual entries, including those that consider inflation and growth of the ABC.

Period	Y	M	D	Segment	MM Demand (Mil $)	Percent Peg	Dividend Percent	Retail Supply Shock(+) Demand Shock(-)	MM Supply Shock(+) Demand Shock(-)	Buy and Destroy Elsies demand shock (-)	Rescue Mode
800	2027	3	20	A	$31.46	99.00%	7.00%	0	0	0.00	$0
800	2027	3	20	B	$22.78	99.00%	7.00%	0	0	0.00	$0
800	2027	3	20	C	$33.00	99.00%	7.00%	0	0	0.00	$0
800	2027	3	20	D	$33.00	99.00%	7.00%	0	0	0.00	$0
800	2027	3	20	E	$33.11	99.00%	7.00%	0	0	0.00	$0
801	2027	3	21	A	$33.11	99.00%	7.00%	0	0	0.00	$0
801	2027	3	21	B	$23.66	99.00%	7.00%	0	0	0.00	$10

The entry in 801-B is $10 million of rescue purchases. This repeats every ten days.

Rescued properties pay no rent, so they typically treble at once (for $1). The philanthropist gets 50% of the Elsies minted and 50% of the structure and premium on treble. Tax breaks could allow the philanthropist to break even or profit from the rescue. Candidates for rescuers, beyond philanthropists, are large industries leaving an area as a gesture of goodwill, local lenders fearing a real estate market collapse, and county boards fearing blight.

Ongoing Retail Trade

=IF ($E260="A", M259 + (Q260 - Q255) * retail_demand, M259)

Ongoing retail trade is another method of representing the retail holding of Elsies. In this case, the current retail trade holdings are a linear function (**retail_demand**) of the property value this day(**Q260**) minus the property value on the previous day (**Q255**), added to the previous day's total (**M259**).

Retail trade includes those holding Elsies for any reason outside of market making and natural dividend demand, which is much lower than temporary dividend demand. Ongoing retail trade can be used with manual one-time purchases to model cases.

Values of retail_demand above 70% will terminate Phase I as fast as logistics allow, even if the discrete spreadsheet shows delays that will not exist in the continuous implementation.

retail_demand is set to zero when modeling natural demand.

Ram and Jam Contracts Signed

=IF (G260 < 99%, 0, IF (AND ($E260="A", F260 > S260 * 2, O259 > 2.5 * S259), MAX (MIN (F260, (O259 - S260) / (S260 * (1 - MM_Bottom))), 0), 0))

This column is an artifact of the simulation. Ram and jam is the most efficient way to handle demand, and any attempt at a formula will be somewhat inefficient, increasing Phase I duration over implementation results.

The first test (**G260 < 99%**) checks if the percent of the peg is less than 99%. If so, no contracts can be signed. In an implementation, ram and jam is always attempted with incoming land fund Elsies. Either there is success above 99%, or they are destroyed.

Ram and jam contracts are only calculated and signed in segment A (which comprises the bulk of the period). The formula assumes a ram and jam run, where contracts are signed, Elsies minted and sold, new contracts are signed, and so on, until the limit specified in the above formula is reached.

A necessary condition is that demand is greater than the average property price (**S260**). In this case, twice the property value ensures a multiplier of at least 2 (**F260 > S260 * 2**) In the implementation, the ram and jam of Elsies can terminate at any time without a property purchase. Another condition is that the land fund is at least 2.5 times the average property price (**O259 > 2.5 * S259**). If these conditions are met, the ram and jam size is the lowest of the demand itself (**F260**), and the land fund required for the ram and jam run (**(O259 - S260) / (S260 * (1 - MM_Bottom))**), but at least zero.

The available land fund is found by subtracting the average price of a home (**O259 - S260**) from the land fund, as the full price must be paid for the first property. Each additional property costs .0095% times the average property price (**S260 * (1 - MM_Bottom)**).

If the available land fund is negative, or there is no demand, no ram and jam is performed.

Land Fund (U.S. Dollars)

=O259 – IF (AND ($E260="A", N260 > 0), S260 + (N260 - S260) * (1 - MM_Bottom), 0) + IF (AND ($E260="E", F259 > Loan_Criterion, O259 < 2 * S259), 1, 0) + IF (AND ($E260="A", P259 < 0, O259 > 1), - 1, 0) + IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / 99.05)), AH259 * MM_Bottom)) – IF (AND (E260="A", I260 < 0), (S260 - S260 * MM_Bottom + ((-I260 - S260) * (1-MM_Bottom))))

In the implementation, the land fund is distributed from the rents and auction proceeds in whatever currency is left over after currency requirements for other recipients have been met. The land fund will typically be distributed in Elsies from rents and dollars from auction proceeds. In the simulation, the land fund always receives Elsies from rents and auction proceeds (auction proceeds are converted to Elsies through the market maker). Although this is more expensive than pure ram and jam, it comes closest to simulating the implementation in a declarative spreadsheet.

Period	Segment	Land Fund (Mil $)	Land Fund Loan
0	E	$1.00	
1	A	$0.64	0
1	B	$0.64	0
1	C	$0.64	0
1	D	$0.64	0
1	E	$0.64	0

The U.S. dollar land fund pays the total price for the first property in ram and jam and 0.0095% for each property after that, assuming a single average property price during a theoretical single run of ram and jam. Property prices will vary in implementation, and multiple runs of ram and jam will merge and diverge. Still, in this simulation, a single run captures all Elsie demand, provided the land fund is sufficient to support the entire run. Otherwise, it runs to the extent of the land fund.

Subtracted from the current land fund (**O259**) in segment A is the land fund cost of running ram and jam (**S260 + (N260 - S260) * (1 - MM_Bottom)**). The average property price (**S260**) plus additional contracted properties in the run (**N260 - S260**) multiplied by the land fund cost of subsequent properties in a ram and jam, .0095% of the purchase price (**1-MM_Bottom**). In an implementation, ram and jam captures unseen demand below 99.05% of the peg down to 99.01% of the peg, which is not included in the simulation.

The two operations that follow (**IF (AND ($E260="E", F259 > Loan_Criterion, O259 < 2 * S259), 1, 0) + IF (AND ($E260="A", P259 < 0, O259 > 1), - 1, 0)**) handle the case where demand for Elsies is strong enough to trigger a large run of ram and jam, but the land fund is insufficient to handle the demand. A loan is taken out in Segment E to cover the shortfall. If funds exist, the loan is repaid in segment A. A meager land fund can support a large ram and jam. While a land fund equal to the price of an average property can only purchase one property, a land fund twice that size can buy 51 properties. On closing, these 51 properties will generate far more money for the land fund than needed to trigger the action.

This operation checks if borrowing into the land fund is profitable, to be repaid at property closing (**F259 > Loan_Criterion, O259 < 2 * S259**). If the demand for Elsies (**F259**) is greater than a set parameter (**Loan_Criterion**) (set in the simulations to $4 million) and the land fund is less than twice the average property price (**O259 < 2 * S259**), then borrow $1 million into the land fund to be repaid after closing, if not before. The ABC market maker will usually make the loan interest-free. This condition is far more likely to occur in a discrete spreadsheet than in an implementation. Land fund dollars are never converted to Elsies in an implementation. They are instead used to purchase properties in the absence of Elsie demand. Long runs of increasing loans in a simulation are not the red flags they appear to be. Good management prevents this situation from happening more than once or twice early in Phase I.

The repayment operation happens in segment A. If the land fund loan column shows that money is owed (**P259 < 0**), and if the land fund has more than a million dollars (**O259 > 1**), repay the million dollars.

The next operation also happens in segment A (**IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / 99.05)), AH259 * MM_Bottom))**). If the land fund is less than current demand (**O259 < F259**), and either the land fund is less than $4 million (**O259 < 4**), or the land fund is less than the amount needed to fund a run of ram and jam for all the demand (**O259 < (F259 * S259) / 99.05**), then convert the Elsie balance to U.S. dollars at the market for use in purchasing land (**AH259 * MM_Bottom**). This operation will never or rarely occur in an implementation. The auction dollars for the land fund would never have been converted to Elsies in the first place.

The final operation also happens in segment A. It is like the first operation, except that instead of buying properties with dollar demand, it is buying properties with

Elsie demand (**IF (AND (E260="A", I260 < 0), (S260 - S260 * MM_Bottom + ((-I260 - S260) * (1-MM_Bottom))))**). These are demand shocks (**I260 < 0**) in Elsies (indicated by a negative sign and in segment A only), where the purchaser buys Elsies from the ABC during ram and jam rather than the market maker. The first hit on the land fund is (**S260**), as in the dollar case. As in the dollar case, the run extracts (1 - MM_Bottom) from the land fund for the remaining properties (**-I260 - S260**). The difference comes with the money left over at the end (**S260 * MM_Bottom**), which must be added back to the land fund.

Land Fund Loan Account

=P259 + IF (AND ($E260="E", F259 > Loan_Criterion, O259 < 2 * S259), -1, 0) + IF (AND ($E260="A", P259 < 0, O259 > 1), 1, 0)

The loan account inherits its previous value in column P and has two operations.

These two operations must have identical conditions to the corresponding operations in the land fund. The decision of when to make a loan and when to repay the loan must be the same in both columns, or the spreadsheet will not balance.

The first operation makes the loan, so it is -1 in the loan account and +1 in the land fund. Repayment of the loan is +1 in the loan account and -1 in the land fund.

This interest-free loan is an artifact of the simulation and will not likely occur in a ram and jam implementation.

Period	Y	M	D	Segment	MM Demand (Mil $)	Ram and Jam Contracts Signed	Land Fund (Mil $)	Land Fund Loan
2387	2031	8	17	E	$703.70	$0.00	$5.42	0
2388	2031	8	18	A	$703.70	$703.70	$1.55	0
2388	2031	8	18	B	$503.92	$0.00	$1.55	0
2388	2031	8	18	C	$726.02	$0.00	$1.55	0
2388	2031	8	18	D	$726.02	$0.00	$1.55	0
2388	2031	8	18	E	$724.93	$0.00	$0.55	1
2389	2031	8	19	A	$724.93	$0.00	$91.00	1
2389	2031	8	19	B	$717.48	$0.00	$91.00	1
2389	2031	8	19	C	$878.79	$0.00	$91.00	1
2389	2031	8	19	D	$939.68	$0.00	$91.00	1
2389	2031	8	19	E	$939.68	$0.00	$91.00	1
2390	2031	8	20	A	$939.68	$939.68	$80.79	0
2390	2031	8	20	B	$703.76	$0.00	$80.79	0

With $725 million in demand and a land fund deficit of $1.55 million, a loan is needed to complete ram and jam. Even though there is still a deficit, spreadsheet internals see that funds will be available before ram and jam is complete, so no additional loan is taken. This triggers a ram and jam run for over 3,000 properties (not shown). Because actual properties vary in price, fractional properties can be purchased to maintain spreadsheet balance. The meaning of a negative (red) land fund is that the loan was too low. This column is more of a check for the

spreadsheet designer. Unpaid loans, or deep nesting of loans, could indicate a logic error; otherwise, short runs of negative balances can be ignored.

Property Value

=Q259 * IF ($E260="A", (1+Commons_Property_Appreciation / (12 * Periods_Per_Month)), 1) + N260 + L260 + IF (AND (E260="A", I260 < 0), -I260 / MM_Bottom)

The property value column is the previous property value(**Q259**) adjusted for daily property appreciation in segment A and totaled with new ram and jam contracts signed(**N260**) and purchases in rescue mode(**L260**). **Periods_Per_Month** is 30, so daily property appreciation is 1/360 of **Commons_Property_Appreciation**.

Added to property value is the property purchased by the ABC during ram and jam to satisfy significant Elsie demand (**I260**). The dollar value of property purchased in this manner is (**-I260 / MM_Bottom**).

With its commons trust property appreciation adjustment, property value determines the rent people will voluntarily pay and the cost of trebling property. It is used to estimate voluntary rent, retail as a percent of property value, the amount of Elsies sequestered in treble escrow, and the degree to which Elsie is backed by land.

General Inflation rate		2%	Commons Property Appreciation	4%

The general inflation rate and commons trust property appreciation are separate parameters. Property appreciation is generally higher by 2% or 3% over inflation. In testing the megacity concept, this difference could be much higher.

Total Purchases (Number of Properties)

=R259 + (N260 + L260 + IF (AND (E260="A", I260<0), -I260 / MM_Bottom))/ S260

The number of properties purchased shows the number of properties, at the average property price, that have been purchased. As the average property price rises with inflation, the number of properties purchased at the average price drops for identical contract amounts. The previous number of properties

purchased(**R259**) is added to the sum of ram and jam contracts(**N260**) and rescue mode purchases (**L260**).

Also added to the total are properties purchased with Elsie demand that go directly to the ABC during ram and jam (**-I260**). Elsie demand has a negative sign and must be in segment A if it is a ram and jam rather than a market bid. The dollar value of properties purchased with Elsie demand is
(**-I260 / MM_Bottom**).

The additions are then divided by the average cost per property (**S260**) to give the number of properties added.

This number is inflation-independent and represents the same chunk of the Earth, regardless of the duration or price appreciation during Phase I. In the implementation, there are multiple criteria for ending Phase I. However, this is the metric used to compare different tests in the simulation. The end of Phase I is defined in the simulation as 190 million, $250,000 properties (2022 dollars). Although the land is held worldwide, this is well over the total value of U.S. residential property in 2022, or $47.5 trillion. Should Phase I take 80 years or longer, the nominal figure would be in the quadrillions, but the actual chunk of the Earth would be the same.

Because the average price of a property is an average, fractions are maintained in total purchases and elsewhere on the spreadsheet. Discrete average property prices do not require discrete math. Any rounding would contradict a balanced spreadsheet.

Average Property Price

```
=S259 * IF ($E260="A", (1 + General_Inflation_Rate / (12 *
Periods_Per_Month)), 1)
```

All columns except those on either side of Average Property Price + period and segment are hidden. Notice that the average property price is initialized to $250,000. Because the simulation takes place over many years, the average price the ABC pays for property will increase with inflation. This price has only a minor effect on the outcome of the simulation. Primarily, it is used to determine how

much of the land fund must be destroyed (1 property's worth) for a ram and jam session.

The value from the previous month (**S259**) is multiplied by one plus the daily rate of inflation (**General_Inflation_Rate / (12 * Periods_Per_Month)**).

Market Maker Elsie Account

= T259 – IFERROR (IF ($E260="B", 2.49% * Q110 / (12 * Periods_Per_Month), 0), 0) * 99% / G260 – IF ($E260="B", 57% * 95% * AR260 / MM_Top, 0) + IF (AND ($E260="A", N260 > 0), (N260 - S260), 0) - IF (E260="D", W260 / MM_Top, 0) + IF ($E260="D", V260, 0) - (AF260 - AF259) + IFERROR (IF ($E260="A", AI259 * T257 / (T257 + AD257)), 0) – IF ($E260="A", (Q260 - Q255) * retail_demand / MM_Top) + IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / MM_Bottom)), AH259)) + IF (E260="E", I260) + IF(E260="E", K260, 0)

The market maker is at the center of all currency conversion in the simulation.

The market maker has two inventories: an inventory of U.S. dollars and an inventory of Elsies. This account is the inventory of Elsies. As such, it is a red-headed column.

In an implementation, the Elsie inventory can become depleted, but in the simulation, temporary depletion of the Elsie inventory is depicted as a negative number. A negative number in this column means the purchaser needed to use U.S. dollars or persuade a retail Elsie holder to sell them Elsies below 100% peg.

The ABC is the primary market maker. Market making is profitable, so there will be competition. Although multiple market makers lower the profits, they speed up Phase I. The highest profits occur (about a 36% return) when the inventory is the size that turns over daily. In the early days, the ABC might do ram and jam once a month or less. Nobody but the ABC will be the market maker initially because even small inventories will be idle. As natural demand increases the rate of purchases, the need for more market-making inventory increases, which speeds up purchases even faster. The ABC market-making inventory is an investment, as needed, from ABC's profits. If ABC profits are insufficient and others do not step up, some property owners and treblers will be forced to use U.S. dollars for transactions. This condition shows up as a negative number in this column.

Every operation in the Elsie inventory requires an “almost equal” and opposite transaction in the dollar inventory. The “almost” refers to the spread or source of market maker profits.

From the starting inventory (**T259**), the first operation in segment B subtracts Elsies for the rent to be paid to the advanced rent fund (**IFERROR (IF ($E260="B", 2.49% * Q110 / (12 * Periods_Per_Month), 0), 0) * 99% / G260**) 2.49% of the previous month's property value (**Q110**) divided by 360 (days in a simulation year).

The multiplier (**99%/ G260**) is typically 1 when G260, the current percent of the peg, is 99%. In the event of deep discounting, rents rise by this amount (if they do not, treblers will push them up even higher).

The “almost equal and opposite” transaction was the property owner's payment of dollars for these Elsies. “Almost equal” because instead of **-2.49% * Q110/(12…**, the dollar inventory entry, seen later, is **+2.49% * MM_Top * Q110/(12…**. The difference between an operation in the Elsie inventory and dollar inventory will always be either an MM_Top or MM_Bottom, one or the other, in either the numerator or denominator. If one is negative, the other will be positive, and vice versa.

The next operation, also performed in segment B (**IF ($E260="B", 57% * 95% * AR260 / MM_Top, 0)**), subtracts Elsies that originated from auction proceeds (**AR260**) and are headed to the Elsie distributor. The percentage of Elsies from the auction proceeds converted from dollars is 57%. This is inefficient but reasonable since most land funds are destroyed. However, for those that make it into the actual land fund, conversion into Elsies, only to be converted back to dollars during ram and jam, will show a slower simulation versus the implementation. The result is multiplied by 95% of the auction proceeds and divided by MM_Top. Recall that 5% of the auction proceeds go to the advance rent fund and are not part of this distribution.

Why does MM_Top appear in the denominator of the Elsie inventory operation for this operation and in the numerator of the dollar inventory operation for the previous operation? In that operation, the number of Elsies needed was known. The dollars input was based on the known number of Elsies to be output. In this operation, the dollars from auction proceeds are known. The Elsie output is based on the known size of the dollar input.

The next operation occurs in segment A (**IF (AND ($E260="A", N260 > 0), (N260 - S260), 0)**) in the event of ram and jam. Elsies created during ram and jam (**N260**) are input into the Elsie inventory. These Elsies are converted to dollars to purchase additional properties. The Elsies converted are equal to the size of the ram and jam minus an average property price (**S260**). A final property's worth is destroyed due to a lack of demand. In the case of the simulation, the market maker purchases Elsies at 99.05% until its Elsie inventory is back to the desired size. When finished, it leaves the ABC holding one average property price worth of Elsies that must be destroyed. Reality is far more complex, with sales down to 99.01%, multiple bidders, properties of various prices, and the ABC stopping ram and jam early. However, all of these make the simulation the conservative option.

The next operation occurs in segment D (**IF (E260="D", W260 / MM_Top, 0)**). (**W260**) holds rent from the dollar advance rent fund that must be converted to Elsies for distribution. These Elsies are leaving for the Elsie distributor, so the sign is a minus. The dollar amount is the known quantity, so the MM_Top appears in the denominator of the Elsie inventory operation. Rent is a confusing term here. The first operation was monthly rent from property owners, who converted their dollars to Elsies to save 0.85% from the arbitrage. This operation is rent from the dollar advance rent funds that must be converted to Elsies for distribution to Elsie recipients. This operation only occurs in the early months of the simulation, where the advance rent funds in dollars funded from auction proceeds exceed the size of the advance rent funds in Elsies from ordinary rent payments.

The next operation is an acquisition (+) rather than a disbursal (-) of Elsies (**IF ($E260="D", V260, 0)**). Segment D (**V260**) holds the Elsies to be converted to dollars. These Elsies come into the Elsie market maker from the Elsie distributor. Dollars will leave from the dollar market maker for the dollar distributor. Because the number of Elsies is known, the MM_Bottom will appear in the denominator of the operation of the dollar market maker.

The operation that follows applies to all segments (**AF260 - AF259**). The process disburses Elsies to sequester in a treble escrow account. Rather than calculating, this operation disburses the difference between the amount in treble escrow in the previous segment and the amount in the current segment. The number of Elsies is known from the treble arbitrage formula, so the MM_Top appears in the numerator of the dollar market maker operation.

The next operation is not a market-maker operation (**IFERROR (IF ($E260="A", AI259 * T257 / (T257 + AD257)), 0)**). Instead, it is the dividend on all Elsies held by the market maker in segment C of the preceding period. The only Elsies that receive dividends are those held by the market maker and retail Elsies. The ratio is the proportion of market maker Elsies(**T257**) to the total of market maker and retail Elsies (**T257 + AD257**). (**AI259**) is the dividends payable. A bug in the spreadsheet increases the size of the dividend for the retail sector when the market maker has a negative inventory. Everything balances because the market maker “contributes” this extra dividend. The result is not material to the simulation.

The next market maker operation (**IF ($E260="A", (Q260 - Q255) * retail_demand / MM_Top)**) is converting retail dollars to Elsies. This is the retail sector's size change from the previous month. It is the change in total property value (**Q260 - Q255**) times the parameter “Retail/Savings Percent of Property Value.” The MM_Top is in the denominator here because dollars, not Elsies, are the known quantity.

The next market maker operation (**IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / MM_Bottom)), AH259)**) indicates whether the land fund in Elsies should go to the actual land fund (converted to dollars). Alternatively, they are destroyed. Land fund Elsies (**AC15**) are needed in the land fund if the land fund is less than the demand for Elsies, and either the land fund is less than $4 million, or the land fund is insufficient to handle a complete ram and jam run for the current demand. Otherwise, the Elsies will be destroyed. In the implementation, Elsies can be split between the land fund and the garbage pail on any given day. In the simulation, an entire month’s worth of Elsies must go one place or the other. This, in addition to the conversion of land fund dollars to Elsies and back again, marks an inefficiency that could see significant improvement in the implementation. The formula here must be identical to the formula in the land fund (save details that determine the market maker spread).

The next operation (+**IF (E260="E", I260)**) is the acquisition or disposition of Elsies from a supply or demand shock. Only in segment E of (**I260**) are the Elsies handled by the market maker. This is treated as an addition of Elsies since the sign of I260 is inverted. The dollar market maker must know if this was a supply shock (+) or demand shock (-) to decide between MM_Top and MM_Bottom.

The final operation (**IF (E260="E", K260, 0)**) removes Elsies purchased in a game or by a generous donor. These Elsies will be destroyed.

MM Desired Inventory

=MAX (U259, IF (AND (T259 < 0, T258 > 0), U259 - T259)) - J259 + IF (AND (E260="B", T259 < 0), -T259)

Estimating desired market maker inventory is difficult when the optimal strategy depends on the source of funds. In the implementation, this is a judgment call on the part of the market maker. If the inventory is too small, it will frequently become depleted (negative), and opportunities will be missed. If the inventory is too large, profit margins will be small.

Sudden purchases lead to large swings in the inventory, which can be used as a trigger. Slow, steady inventory investments leave a different set of evidence. Readers are encouraged to seek better formulas (and let me know if you find something promising).

If the inventory shifts from positive to negative (**AND (T259 < 0, T258 > 0)**), set the desired inventory to the maximum of the current desired inventory and the difference between the current desired inventory and the current inventory (**U259 - T259**), which is negative.

Users can adjust inventory from the MM demand and supply shocks columns (**J259**). This is different from its neighboring retail demand shocks and supply shocks column. These adjustments are added or subtracted to/from the desired inventory.

If the inventory in segment B is negative, then add the absolute value of that deficit to the desired inventory (**IF (AND (E260="B", T259 < 0), -T259)**).

Elsie Rent to be Converted to Dollars

=IF (E260="E", 0, V259 + IF (E260="D", IF (AB259 > 57% * (Y259 + AB259), AB259 - 57% * (Y259 + AB259), 0), 0))

Period	Y	M	D	Segment	Elsie Rent to be converted to dollars (I)	Dollar rent to be converted to Elsies (H)
657	2026	10	27	A	0.00	0.00
657	2026	10	27	B	0.00	0.00
657	2026	10	27	C	0.00	0.00
657	2026	10	27	D	0.00	0.12
657	2026	10	27	E	0.00	0.00
658	2026	10	28	A	0.00	0.00
658	2026	10	28	B	0.00	0.00
658	2026	10	28	C	0.00	0.00
658	2026	10	28	D	0.00	0.12
658	2026	10	28	E	0.00	0.00
1668	2029	8	18	A	0.00	0.00
1668	2029	8	18	B	0.00	0.00
1668	2029	8	18	C	0.00	0.00
1668	2029	8	18	D	0.19	0.00
1668	2029	8	18	E	0.00	0.00
1669	2029	8	19	A	0.00	0.00
1669	2029	8	19	B	0.00	0.00
1669	2029	8	19	C	0.00	0.00
1669	2029	8	19	D	0.19	0.00
1669	2029	8	19	E	0.00	0.00

In segment D, the rent in Elsies (**AB259**) is checked to see if it is greater than the 57% of the total rent needed to be distributed as Elsies (EDSF 25%, Land Fund 25%, Dividend 7% = 57%). Total rent is the sum of dollar rent and Elsie rent (**Y259 + AB259**). If the rent in Elsies is greater than the Elsies needed by recipients, the remainder (**AB259 - 56% * (Y259 + AB259)**) is sent off to be converted to dollars.

This column will be zero if the U.S. dollar rent contribution of new purchases exceeds the Elsie rent contribution of properties already in the commons trust. It is zero throughout the first year in the 20-year Phase I simulation. It becomes non-zero in January 2027. Note that this and the column Dollar Rent to be converted to Elsies are mutually exclusive since there can only be a surplus in one direction. Both cannot have non-zero values at the same time.

Dollar Rent to be Converted to Elsies

=IF (E261="D", IF (Y260 > 43% * (Y260 + AB260), Y260 - 43% * (Y260 + AB260), 0), 0)

In segment D, the dollar rent in (**Y260**) is checked to see if it is greater than 43% of the total rent needed for dollars. Total rent is the sum of dollar rent and Elsie rent (**Y260 + AB260**). If there are more dollars than required rent needed in dollars, the rest (**Y260 - 43% * (Y260 + AB260)**) are to be converted to Elsies.

Advance Rent Fund (U.S. Dollars)

=X260 + IF ($E261="B", 5% * AR261, 0) - IF (E261="C", X260 / (12 * Periods_Per_Month), 0)

Period	Y	M	D	Segment	Advance rent funds (Mil dollars)	Dollar Rent	Dollar Distributor
1668	2029	8	18	A	$908.11	$0.00	$0.00
1668	2029	8	18	B	$912.56	$0.00	$36.33
1668	2029	8	18	C	$910.02	$2.53	$36.33
1668	2029	8	18	D	$910.02	$0.00	$39.05
1668	2029	8	18	E	$910.02	$0.00	$0.00
1669	2029	8	19	A	$910.02	$0.00	$0.00
1669	2029	8	19	B	$914.41	$0.00	$35.82
1669	2029	8	19	C	$911.87	$2.54	$35.82
1669	2029	8	19	D	$911.87	$0.00	$38.55
1669	2029	8	19	E	$911.87	$0.00	$0.00
1670	2029	8	20	A	$911.87	$0.00	$0.00
1670	2029	8	20	B	$915.52	$0.00	$29.82
1670	2029	8	20	C	$912.97	$2.54	$29.82
1670	2029	8	20	D	$912.97	$0.00	$32.55
1670	2029	8	20	E	$912.97	$0.00	$0.00

The U.S. dollar advance rent fund is funded with rents paid in U.S. dollars and 5% of auction proceeds paid in U.S. dollars. The 20-year Phase I simulation assumes all rent is paid in Elsies and most auction bids until the last few years are in U.S. dollars, which, for the most part, is expected to be true. The dollar advance rent account is funded only with 5% of the auction proceeds (**AR261**) in segment B. If a column for rent paid in dollars is added to the spreadsheet, that would also need to be added here.

In segment C, 1/360 of the fund's current value is removed for distribution. Dollar rents and Elsie rents distributed in the same period are comingled. If the dollar rents exceed 44% of the total, the excess is sent to the market maker to be converted to Elsies. If Elsie rents exceed dollars (which is typical), excess Elsies are sent to the market maker for conversion. In an implementation, this must be done on a county-by-county basis.

Note that this is a purple-headed column that contributes to the total of all dollar accounts.

Dollar Rent

=IF ($E260="C", X259 / (12 * Periods_Per_Month), 0)

Dollar Rent is a temporary holder of U.S. dollars. 1/360 of the U.S. Dollar advance rent fund(**X259**) is moved here in segment C and out in segment D, returning the field to zero. It does hold the dollars for that one segment and must be a purple-headed column to maintain spreadsheet balance.

Dollar Distributor

=IF ($E260="E", 0, Z259 + IF (E260="B", 43% * 95% * AR260, 0) + IF ($E260="D", V260 * MM_Bottom) + IF ($E260="D", Y259 - W260)) + IF (AND ($E260="C", T259 < 0, AE259 > 0), MIN (-T259, AE259) * 0.9916)

The dollar distributor distributes rent and auction proceeds to those recipients who expect dollars. These are the ABC, VTLM, and counties.

The distributor distributes its entire contents in segment E and is reset to zero. Otherwise, operations work on the current value of the distributor(**Z259**). If the segment is B, 43% of the 95% U.S. dollar auction proceeds treated as rent are moved to the distributor (**43% * 95% * AR260**).

In segment D, the distributor gets dollars from the market maker. The market maker converted rent paid in Elsies to dollars for dollar recipients. The market maker purchases the known number of Elsies in (**V260**) at MM_Bottom, the number of dollars moved into the distributor.

The dollar rent (**Y259**) is moved into the distributor in segment D. But not all. Some dollar rent, particularly in the early days when dollars predominate, must be converted to Elsies for the EDSF and dividend. The number of dollars to be converted is found in (**W260**).

When Elsies are purchased in the 99.16% inventory, dollars received go directly to the distributor for delayed distribution. In segment C, dollars are taken from delayed distribution when market maker inventory is depleted (**T259 < 0**), and there are Elsies in the 99.16% inventory (**AE259 > 0**). The extent of the sale is the lesser of the market maker deficit or the size of the 99.16% inventory (**MIN (-T259, AE259) * 0.9916**) multiplied by the price paid $0.9916/Elsie.

Advance Rent Fund (Elsies)

=AA259 + IF ($E260="C", AS259) – IF ($E260="B", AA259 / (12 * Periods_Per_Month)) + IF ($E260="D", (5% * N107 * probability_elsie_bid_win)/2, 0)

The advance rent fund (or account) is a repository of Elsies used to pay rents according to the distribution charter of the commons trust. It is a collection of individual advance rent accounts, one for each property. However, the distributive property of addition allows the accounts to be combined to pay rent. That is, 1/360 of each advance rent account equals 1/360 of their sum.

There is another advance rent fund for dollars because rent can be paid in either Elsies or U.S. dollars. The distributive property works across currencies as well. The monthly rent is 1/360 of the combined advance rent accounts holding Elsies and 1/360 of those containing dollars.

The advance rent fund is an Elsie account summed into total accounts, as seen by its red heading. It consists of the current value of the advance rent fund (**AA259**) plus two operations. The operation in segment C adds the rent paid by property owners to the advance rent fund (**+ IF ($E260="C", AS259)**). However, segment B's operation subtracts 1/360 of the advance rent fund to pay the daily rent. Note that the rent is removed before it is added. In an individual advance rent account, the payments are monthly, not daily. If an individual pays monthly rent in segment C equal to the monthly rent removed in segment B, that individual's advance rent fund and monthly rent remain constant over time.

Auction winning bids could be in Elsies. If so, 5% of the winning bid is transferred to the Elsie advance rent fund in segment D (**+ IF ($E260="D", (5% * N107 * percent_elsie_bid_win)/2, 0)**). The amount is computed from scratch. (**N107**) are the ram and jam proceeds from a month earlier. (**percent_elsie_bid_win, T7**) is the parameter “Percent of Winning Bids in Elsies.” Because 50% of the purchase price is expected for auction proceeds, the product is divided by 2.

Elsie Rent

```
=IF ($E260="D", 0, AB259 + IF ($E260="B", AA259 / (12 * Periods_Per_Month), 0))
```

In segment B, Elsie rent is the daily rent paid from 1/360 of the Elsie advance rent fund (AA259). However, all this fund has been moved elsewhere, in segments C and D, leaving a zero balance in segment D.

In Segment D, what remains goes to the Elsie Distributor. However, this is handled in the Elsie rent column by zeroing out the field.

Even though the “Elsie rent” column does not maintain a balance through all segments, it is a repository of Elsies for a period. It contributes to total accounts, as indicated by its red heading. Funds sequestered here, as in the advance rent fund, are not eligible for dividends.

Elsie Distributor

= IF ($E260="E", 0, AC259 + IF (E260="D", AB259, 0) + IF (E260="B", 95% * 57% * AR260 / MM_Top, 0) + IF (E260="D", W260 / MM_Top, 0)) – IF ($E260="D", V260, 0)

The Elise Distributor distributes Elsies acquired from rent paid in Elsies, auction proceeds paid in Elsies, or rent/auction dollars converted to Elsies by the market maker to recipients of Elsies as set in the commons trust charter. These are the EDSF, dividend payable, and land fund. However, Elsies, not needed by the land fund, are destroyed rather than distributed.

It is a waste to convert dollars from property auctions into Elsies only to destroy or convert them back to dollars. However, the logic of the simulation would be too complex to handle alternate cases. In an implementation, land fund dollars might be used to purchase property if the land fund is overflowing and the market maker's inventory of Elsies is somewhat depleted.

In the implementation, Elsies received in the land fund are destroyed or converted to dollars through ram and jam.

All Elsies are distributed in segment E. This is handled by zeroing out the distributor in segment E.

The first operation shown happens in segment D, where the entirety of the "Elsie rent" column (**AB259**) that was not sent to the market maker for conversion is added to the current value of the distributor (**AC259**).

The distributor also distributes auction proceeds. This operation occurs in segment B. Recall that 5% of auction proceeds are used to initialize an advance rent account, while the other 95% are distributed as rent. Typically, auction proceeds will be in dollars, particularly in the early days of the ABC. All auction proceeds in the (**AR260**) column are in dollars. Auction proceeds in Elsies are excluded from the column.

If all the land fund is to be kept in Elsies, the percent of rent and auction proceeds that must be held in the Elsie distributor is 57% (25% land fund, 25% EDSF, and 7% dividend). These Elsies will typically come from the market maker and always come from the market maker in the simulation. The number of Elsies is (**95% * 57% * AR260 / MM_Top**). Where (**AR260**) is the auction proceeds, where MM_Top is .9915 dollars/Elsie.

In segment D, the distributor receives rent dollars that the market maker must convert to Elsies (**W260**). This is typically zero, as rent is always paid in Elsies in the simulation. However, the 5% of auction proceeds that go to the advance rent fund as dollars will cause the value to be non-zero when purchases have a greater weight than properties in the commons trust.

The final operation in segment D, before distribution in segment E, removes those Elsies to be converted to U.S. dollars (**– IF ($E260="D", V260, 0)**) through the market maker.

Elsie Distributor is a red-headed column, holding Elsies in segments B, C, and D.

Retail Elsies

=AD259 + IF ($E260="B", L260/2) + IF ($E260="A", (Q260 - Q255) * retail_demand / MM_Top) + IFERROR (IF ($E260="A", AI259 * AD257 / (T257+AD257)), 0) – IF (E260="E", I260, 0) – IF (E260="A", I260, 0) + IF (AND ($E260="C", T259 < 0, AE259 > 0), MIN (-T259, AE259)) – IF ($E260="D", (N107 * probability_elsie_bid_win)/2,0) – IF (E260="B", (N109 * elsie_bids_per_property), 0) + IF (E260="D", (N107 * elsie_bids_per_property), 0)

These are the Elsies in general circulation. The category includes Elsies held by merchants, suppliers, and consumers for trade, as well as Elsies held by investors for dividends, appreciation, or to accelerate the movement to land-based capitalism.

It is comprised of multiple inputs. The first input comes from rescues (**L260**). Recall that 50% of the Elsies minted for a rescue goes to the rescuer (**L260/2**).

The next input is the growing retail trade, which includes all purchases of Elsies into the retail segment not otherwise covered in this formula. It adds the increase in property value since the previous day multiplied by the “Retail/Savers Percentage” parameter (**(Q260 - Q255) * retail_demand / MM_Top**). The product is divided by MM_Top since this is a known quantity of dollars through the market maker.

The next input (**IFERROR (IF ($E260="A", AI259 * AD257 / (T257+AD257)), 0)**) comes from dividends payable (**AI259**). Retail Elsies and Elsies held by the market maker in segment C of the previous day are eligible for the dividend. Although the dividend is paid in segment A, eligibility is determined in the last period’s segment

C. Elsie Rent and Elsie Distributor columns, both artifacts of the simulation, are, like the advance rent fund from which they came, not eligible. Only market maker Elsies in (**T257**) and retail Elsies in (**AD257**) qualify. The dividend is a function of dividends payable multiplied by the proportion of eligible Elsies in retail. The IFERROR substitutes 0 when a divide-by-zero error occurs where the market maker has yet to acquire Elsies, and the simulation does not include a retail segment.

The following two operations handle demand and supply shocks (**– IF (E260="E", I260, 0) – IF (E260="A", I260, 0)**). Retail Elsies are exchanged with the market maker in segment E. Retail Elsies are supplied by the ABC in segment A during ram and jam. Specifying a supply shock (positive value in I260) in segment A is a logic error.

Retail Elsies are purchased from delayed disbursement in the next operation when the market maker inventories are depleted, and 99.16% of inventories exist (**T259 < 0, AE259 > 0**). Elsies purchased are the smallest of either the market maker deficit or the size of the 99.16% inventory (**MIN (-T259, AE259)**).

In the next operation, Elsies bid for the property at auctions comes from retail in segment D (**– IF ($E260="D", (N107 * percent_elsie_bid_win)/2,0)**). Property under contract from a month earlier(**N107**) times the percent of property bid wins in Elsies (**percent_elsie_bid_win**) divided by two since the auction is expected to bring in 50% of the purchase price.

All Elsie bids are expected to come out of retail for the auction, and the losing bids will return to the retail segment after the auction. This final operation (**– IF (E260="B", (N109 * elsie_bids_per_property), 0) + IF (E260="D", (N107 * elsie_bids_per_property), 0)**) sequesters the Elsies in segment B and returns them to retail in segment D. The auction and dividend qualification occurs in segment C.

Delayed Disbursement

=AE259 + IF ($E260="B", 44% * L260/2, 0) – IF (AND ($E260="C", T259 < 0, AE259 > 0), MIN (-T259, AE259)) + IF ($E260="D", (95% * 43% * N107 * probability_elsie_bid_win) /2, 0)

Delayed disbursement is used on 95% of Elsie auction proceeds, 50% of rescue purchases, and 50% of direct mode purchases. Direct mode purchases are not handled in the simulation.

In the first operation, 43% of the Elsies generated by rescue purchases (**L260/2**) are subject to delayed disbursement in segment B. In rescue purchases, 50% of the Elsies minted (**L260/2**) are subject to immediate-delayed disbursement. The 43% of rents going to the ABC, purchasing agent, VTLM, and counties are delayed.

Delayed distributions are Elsie inventories offered at 99.16% of the peg. Major market makers set MM_Top at 99.15%. Only when their stocks are depleted will the delayed distributions occur. Delayed distribution offerings are first-in, first-out by county, although there is no such distinction in the simulation.

If the market maker inventory is depleted (**T259 < 0**) and there is a positive balance in the 99.16% inventory (**AE259 > 0**), sell as much of the inventory as the market maker deficit will allow (**MIN (-T259, AE259)**), at 99.16%. Dollars received go directly to the dollar distributor.

Period	Y	M	D	Segment	Market Maker (Mil LC$)	Dollar Distributor	Retail Elsies	Delayed Disbursement
2970	2033	3	30	E	569	$0.00	717,259.41	6.45
2971	2033	4	1	A	-3,805	$0.00	718,640.21	6.45
2971	2033	4	1	B	-4,275	$295.30	696,953.68	6.45
2971	2033	4	1	C	-4,275	$301.69	696,960.13	0.00
2971	2033	4	1	D	-4,272	$332.88	718,465.94	73.82
2971	2033	4	1	E	-4,280	$0.00	718,465.94	73.82
2972	2033	4	2	A	-4,339	$0.00	718,635.84	73.82
2972	2033	4	2	B	-4,817	$301.85	696,467.92	73.82
2972	2033	4	2	C	-4,817	$375.06	696,541.74	0.00
2972	2033	4	2	D	-4,814	$406.31	718,524.94	75.46
2972	2033	4	2	E	-4,822	$0.00	718,524.94	75.46
2973	2033	4	3	A	811	$0.00	727,346.12	75.46
2973	2033	4	3	B	325	$307.70	704,749.07	75.46
2973	2033	4	3	C	325	$307.70	704,749.07	75.46
2973	2033	4	3	D	328	$339.00	727,157.81	152.39

In the final operation in segment D, 43% of the auction proceeds received in Elsies (**(95% * 43% * N107 * percent_elsie_bid_win) /2**) are subject to delayed disbursement. This is calculated from scratch from the contracted purchase price a month earlier (**N107**) and the percent of auction wins in Elsies (**percent_elsie_bid_win**). The winning bid is expected to be half the purchase price, dividing the product by 2.

In this table, the market maker suffers a 4.275 billion Elsie deficit on April 1, 2033. Money comes out of delayed disbursement and goes into retail. Shortages continued the next day, again clearing the 99.16% inventory. However, the market maker inventory goes positive on April 3, and the inventory is not cleared. The dollars won at 99.16% go to the Dollar Distributor for the ABC, VTLM, and counties.

Sequestered Treble Arbitrage

=IFERROR (IF (G260 < 99%, 100%, percent_of_treblers) * (Q110 * (4/3)/12 + 1.25% * Q110/4), 0) * (100 - G260 * 100)

If the goal is to pay the lowest rent, the most efficient rental strategy is to allow one's rent to fall until trebled and then match the trebler (or surrender the property). However, for those at the very top, high rents bring aristocracy and bragging rights (in land-based capitalism, rents are the new Rolex). For those who live paycheck to paycheck, lack of liquidity prevents matching the trebler. Some would rather pay monthly rent and avoid the hassle of being trebled every year. Everyone will not use the efficient rental strategy.

While treblers tend to be more sophisticated than property bidders at auction and more invested in the AFFEERCE business plan, there is no assurance that all of them will use Elsies to take advantage of the 0.85% arbitrage discount.

If everybody used the efficient rental strategy, and all treblers used Elsies, the value of the (**percent_of_treblers, Q5**) parameter should be set to 100%. Should the Elsie be deeply discounted (below 99% of the peg), percent_of_treblers is automatically treated as 100% as all treblers will use Elsies, and all properties can fall below the treble danger line. The multiplier is (**IF (G260 < 99%, 100%, percent_of_treblers)**).

Because the trebling of those using the efficient rental strategy occurs about once per year, the amount in escrow is a function of 1/12th of the property value. The multiplier is multiplied by 133% (4/3) of the property value in the previous month. Added to this is 1/12 of the rent, but since the amount must be tripled, it is divided by 4.

Sequestered treble arbitrage improves the dividend by limiting the Elsie pool but does not affect the average rent paid except to enforce it. Natural demand for Elsies is reflected in the increase of this arbitrage pool from month to month.

At the end of the formula, notice that everything is multiplied by ((**100 - G260 * 100)**). G260 is the current percent of the peg, typically 99%. So, under normal circumstances, this evaluates to the multiplicative identity 1. However, this number increases in the event of deep discounting. 98% is 2, 97% is 3, and 96% is 4. The TAD table in The Effect of an Elsie Discount on Treble Arbitrage shows a four multiplier for 40% land share at 96%. The TAD curve is relatively linear above 90%, so this approximation is reasonable.

Sequestered Elsie Bids

=IF (E261="B", (N110 * elsie_bids_per_property), 0) + IF (E261="C", AG260)

This red-headed column holds the Elsies sequestered for bidding at auction in segments B and C. Purchases from a month ago (**N110**) are multiplied by the expected number of bidders in Elsies (**elsie_bids_per_property**). Because bidders are only expected to bid 50% of the purchase price, this and the corresponding entry in retail should be divided by 2.

Land Fund Elsies

=IF ($E261="E", AC260 * 25% / 57%, 0)

The land fund receives 25% of the ground rent. In the simulation, the land fund receives all its revenue in Elsies. If the dollar-based land fund is sufficient, the Elsies are destroyed. If not, the Elsies are converted to the dollar land fund through ram and jam. The implementation is more efficient. Rents distributed to the land fund in dollars always go to the dollar land fund. If the dollar land fund is sufficiently large, these dollars are used directly to purchase property, even if there is no ram and jam demand. Rents distributed to the land fund in Elsies are treated the same way they are in the simulation.

The Elsie Distributor (**AC260**) obtains 57% of the ground rent and auction proceeds as Elsies. 25% of that 57% goes to the land fund in segment E.

Dividend Payable

= IF ($E261="E", AC260 * H261/57%, 0) + IF ($E261="E", (95% * H261 * N107 * probability_elsie_bid_win) / 2, 0)

Period	Y	M	D	Segment	Dividend Payable	EDSF	Total Elsies	Annualized Dividend
2970	2033	3	30	E	63.55	129,816.26	938,074	3.2663%
2971	2033	4	1	A	0.00	129,816.26	939,867	0.0000%
2971	2033	4	1	B	0.00	129,816.26	939,867	0.0000%
2971	2033	4	1	C	0.00	129,816.26	939,867	0.0000%
2971	2033	4	1	D	0.00	129,859.18	939,824	0.0000%
2971	2033	4	1	E	65.58	130,050.49	939,816	3.3876%
2972	2033	4	2	A	0.00	130,050.49	939,816	0.0000%
2972	2033	4	2	B	0.00	130,050.49	939,816	0.0000%
2972	2033	4	2	C	0.00	130,050.49	939,816	0.0000%
2972	2033	4	2	D	0.00	130,094.36	939,772	0.0000%
2972	2033	4	2	E	66.94	130,289.55	939,764	3.4596%
2973	2033	4	3	A	0.00	130,289.55	954,104	0.0000%
2973	2033	4	3	B	0.00	130,289.55	954,104	0.0000%
2973	2033	4	3	C	0.00	130,289.55	954,104	0.0000%
2973	2033	4	3	D	0.00	130,334.27	954,060	0.0000%
2973	2033	4	3	E	68.14	130,532.92	954,052	3.4793%

The dividend payable is typically 7% of the total rents and auction proceeds or 7% out of the 57% of rents and auction proceeds held by the Elsie distributor (**AC260**) for the dividend payable, EDSF, and land fund.

However, should the Elsie trade on the market at less than 99% of peg (at the start of segment C, or using some daily average), the computed dividend percentage is held in (**H261**).

In the event of an auction with an Elsie winning bid, H261 (typically 7%) of 95% of the auction proceeds are distributed as a dividend. Dividends from dollar auction bids go through the market maker and Elsie Distributor.

Dividend payable is a red-headed fund that holds Elsies during segment E. By Segment A of the following period, the dividend payable will be paid, and this column is zeroed out. The two recipient columns, retail holders and the market maker know how to divide the dividend.

EDSF

=AJ260 + IF ($E261="E", AC260 * (32% - H261) / 57%, 0) + IF ($E261="B", 57% * L261/2) + IF ($E261="D", (95% * (32% - H261) * N108 * probability_elsie_bid_win) / 2, 0)

The Earth Dividend Subsidy Fund (EDSF) typically receives 25% of the total rent and auction proceeds or 25% of the percentage held by the Elsie distributor. Strictly speaking, it receives 32% minus the share going to the dividend (**H261**), typically 7% if the Elsie trades at 99.00% of peg or better. Like the dividend payable, it receives its rent payment from the distributor (**AC260**) in segment E. Unlike the dividend payable, the EDSF accumulates throughout Phase I.

The EDSF is funded from Elsies received in rescue mode. Half the Elsies go to the philanthropist, and the other half are distributed as rent in immediate-delayed disbursement. However, the EDSF additionally receives that portion of rent for the dividend and land fund or the complete 56% that is not delayed (**57% * L261/2**).

For auction wins in Elsies, typically 95% of 25% (**(32% - H261)**) of purchase from one month earlier (**N108**) with the winning bid expected to be half (**/ 2**) times the percentage of auction wins in Elsies (**percent_elsie_bid_win**) goes to the EDSF.

Eight years before the end of Phase I, dividends can be awarded to residents of Phase II communities or Earth Dividend auction winners. This is done by transferring the actuarial present value of the Earth Dividend from the EDSF to a present value fund and then distributing the Earth Dividend each month to each recipient. The movement of Elsies out of the EDSF and present value fund and into circulation is tiny compared to the value of those funds and not shown in the simulation. In this simulation, Elsies in the EDSF are permanently sequestered.

Total Accounts

=T260 + SUM (AA260:AJ260) + AS260

This is the total of all the red-headed accounts. To be balanced, the value here must always equal the value in the Total Elsies column. The column is shown in green for easy comparison with the Total Elsies column, which is also in green.

T260	Elsie Market Maker
AA260	Elsie Advance Rent Funds
AB260	Elsie Rent
AC260	Elsie Distributor
AD260	Retail Elsies
AE260	Delayed Disbursement
AF260	Sequestered Elsies for Trebling
AG260	Sequestered Elsies for Bidding
AH260	Land fund Elsies
AI260	Dividend Payable
AJ260	EDSF
AS260	Rent Paid by Property Owners

Elsies Created

=AL259 + N260 + IF ($E260="B", L260) + IF (E260="A", -I260)

This is a running total of all Elsies created. Elsies can only be minted when a property is purchased into the Commons Trust. This occurs through ram and jam reflected in (**N260**) and rescued property purchases, reflected in (**L260**) and recorded in segment B. Property can also be purchased with Elsie demand(**-I260**) directed at the ABC during ram and jam, bypassing the market maker. Demand is negative, requiring the minus sign. Positive supply in segment A is not functional as it implies privatizing commons trust land. There is no provision in either the simulation or implementation to do so.

Elsies Destroyed

=AM259 + IF (AND (N260 > 0, $E260="A"), S260, 0) + IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / MM_Bottom)), 0, AH259)) – IF (E260="E", K260) + IFERROR (IF ($E260="D", (95% * 25% * N107 * probability_elsie_bid_win) / 2, 0), 0)

This is a running total of Elsies destroyed. In version 6.0, Elsies Destroyed was an account called the bank and would have been shown as a red-headed column. In 7.0, destroyed Elsies can never be resurrected. They are subtracted from Elsies created to get the total Elsies.

Every theoretical run of ram and jam destroys an average property price's worth of Elsies. In segment A, in the event of ram and jam (**N260 > 0**), Elsies equal in value to the final property purchased (average property value in simulation) (**S260**) will be destroyed. In other words, ram and jam can go on for an arbitrary number of iterations, each subsidized with 1% of the purchase price from the land fund. However, when demand is finally exhausted, one average property price worth of Elsies must be destroyed.

In the simulation, all land funds not converted to dollars are destroyed (**+ IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) /**

MM_Bottom)), 0, AH259))). In the implementation, all such Elsies are put through ram and jam to either be converted to dollars or destroyed. However, the ABC has the right to save these Elsies until, for instance, there is demand on the market. Or, they can destroy the Elsies, even if there is demand. Dollars received from the sale of Elsies can be used to purchase land at the time or in the future. The simulation is not necessarily optimal.

The condition (**AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) /MM_Bottom**) checks to see if the Elsies are **not to be** destroyed. That way, it matches the state in the land fund where land fund Elsies are converted to dollars, with the if and else clauses reversed. We do not destroy the Elsies if the land fund is less than demand (**O259 < F259**) and either the land fund is less than 4, or the land fund is inadequate to do an entire run of ram and jam with current demand (**O259 < (F259 * S259) / MM_Bottom**). Should the condition fail, the contents of the Elsie land fund (**AH259**) will be destroyed.

If the winning property bidder is in Elsies, all Elsies in the land fund are destroyed (**(95% * 25% * N107 * percent_elsie_bid_win) / 2**). **N107** are the property contracts from a month earlier **percent_elsie_bid_win** is the percent chance the winning bid will be in Elsies. The product is divided by two because only 50% of the purchase price is expected to be bid.

Total Elsies

=AL260 - AM260

Total Elsies is Elsies created (**AL260**) minus Elsies destroyed (**AM260**).

Period	Segment	Total Accounts	Elsies Created	Elsies Destroyed	Total Elsies
0	E				
1	A	11.7496	12.00	0.25	11.7496
1	B	11.7496	12.00	0.25	11.7496
1	C	11.7496	12.00	0.25	11.7496
1	D	11.7496	12.00	0.25	11.7496
1	E	11.7496	12.00	0.25	11.7496
254	A	681.7009	731.71	50.01	681.7009
254	B	696.7009	746.71	50.01	696.7009
254	C	696.7009	746.71	50.01	696.7009
254	D	696.7009	746.71	50.01	696.7009
254	E	696.7009	746.71	50.01	696.7009
255	A	700.2291	750.62	50.39	700.2291
255	B	700.2291	750.62	50.39	700.2291
255	C	700.2291	750.62	50.39	700.2291
255	D	700.2291	750.62	50.39	700.2291
255	E	700.2291	750.62	50.39	700.2291
256	A	704.5420	755.31	50.77	704.5420
256	B	704.5420	755.31	50.77	704.5420
256	C	704.5420	755.31	50.77	704.5420
256	D	704.5420	755.31	50.77	704.5420
256	E	704.5420	755.31	50.77	704.5420

Amazingly, the total Elsies must always be equal to the sum of the Elsies in all the red-headed accounts in the universe. This is called zero-balancing because the difference between these two green columns is zero.

Elsies Earning a Dividend

=IF ($E260="C", MAX (MAX (AD260, 0) + MAX (T260, 0), 0.0001), 0)

The only Elsies that can earn a dividend are those held by the ABC market maker (**T260**) and those held by the public (**AD260**). If no Elsies qualify for the dividend, .0001 (100 Elsies) is used to prevent a divide-by-zero error and prevent the spreadsheet from showing a huge number. Enormous dividend spikes will occur from time to time, but rarely will so few as 100 Elsies remain in circulation.

Should the market maker inventory or retail Elsies go negative, they will not count toward computing the dividend.

Dividends are paid to the owner, on record, of Elsies during segment C.

In the implementation, there can be situations where almost all Elsies are locked away, and those holding Elsies during segment C can earn a lottery-win-sized dividend on their Elsies. The dividend will naturally increase as Phase I progresses unless the retail segment grows faster. Phase II dividends grow extremely fast, triggering an ever-deepening deflation and hyperdeflation.

Annualized Dividend

=IF ($E260="E", (AI260 / AO258) * 12 * Periods_Per_Month, 0)

Dividends are paid once per day. The annualized dividend is computed in segment E after the dividend payable (**AI260**) has been calculated. The column “Elsies earning a dividend” in segment C (**AO258** is divided into the dividend payable column and then multiplied by the number of periods (360) in the year to annualize the result.

Land Backing

=40% * Q110 / AN260

The goal of the ABC is to purchase property with an average land share of 40%. When purchased, Elsies are minted equal to the property value, 40% of which is land. Dividing the total property value (**Q110**) by the total Elsies (**AN260**) gives the property value ratio to Elsies. By multiplying this ratio by 40%, the percentage of Elsies that are entirely backed by land is found.

Factors influencing this column are the number of Elsies destroyed and the property appreciation rate set as a simulation parameter. The property appreciation rate includes inflation and the value of new construction on the property, which is expected to be high due to the lack of property taxes. Depreciation reduces Commons Trust property appreciation.

Rapid purchasing will bring down the number of Elsies fully backed by land (as well as the dividend). Slowing down the rate of purchases will raise both numbers.

Because of the 30-day delay before closing, this number can be less than 40% in the first few years of Phase I.

Auction Proceeds (Dollars)

=IF ($E260="B", (N109 * (1 - probability_elsie_bid_win)) / 2, 0)

The auction proceeds in dollars equal the ram and jam contracts from a month earlier (**N109**) multiplied by one minus the percentage chance the auction proceeds will be in Elsies (**1 - percent_elsie_bid_win**). Because proceeds are expected to be half of the purchase price on average, the product is divided by 2.

The auction proceeds in Elsies do not have their own column. Instead, the formula (**(N109 * percent_elsie_bid_win) / 2**) is used instead of a column reference.

Rent into the ARF

=IF ($E260="B", 2.49% * Q110 / (12 * Periods_Per_Month), 0) * 99% / G260

This is the average rent voluntarily paid by tenants into their advance rent fund to avoid a treble. If the average land share purchased by the ABC is 40%, then 2.49% of the purchase price will be the average rent (See module, Average Rent as a percent of Purchase Price). As total purchase price increases with property appreciation, average rents increase by the same amount. This concept can be confusing, as rent on a particular parcel will drop as the property is developed. However, theory (and empirical data) tells us that land value increases on neighboring lots will more than compensate for the loss of land value on a developed property. If everybody builds, all rents will rise. If that were not true, nobody would pay rent to live in Manhattan.

Notice that even though this is row 260, the property value, **Q110**, is from the previous month. Rent is not paid on properties under contract but on closed properties. It is assumed the closing takes place one month later. The property value column includes properties under contract. It is also reasonable, in general, that the previous month's value will determine the rent paid. (Of course, nobody knows what the rent should be, which is why the treble market exists.)

The product is divided by 360 (**12 * Periods_Per_Month**) to compute daily rent.

All of this is multiplied by (**99%/G260**). Column G is initialized and has a value of 99% during most simulations. However, during simulations of retail Elsie dumps,

the value can drop as low as 93% of the peg. Due to a rash of trebling, rents are forced up, at least temporarily, by the discount. Even when rents subsequently return to their 99% level, funds added to the advance rent funds during the “deep discount” do not return to the property owner, so the deep discount acts as an Elsie sink.

This column, rent in, is an Elsie fund column. Notice that the heading is in red. The red header (a red-headed column) means it is one of the accounts summed in the total account’s column.

Total Dollars Input

=AT259 + AR260 + IFERROR (IF ($E260="B", 2.49% * Q110 / (12 * Periods_Per_Month) * MM_Top, 0) * 99% / G260 + (AF260 - AF259) * MM_Top, 0) + IF ($E260="B", L260) + IF ($E260="A", (Q260 - Q255) * retail_demand) + IF (AV260 < U260, MIN (U260 - AV260, AV260 * 0.1), 0) -IF (E260="E", IF (I260 < 0, I260 * MM_Top, I260 * MM_Bottom)) – IF (AND (E260="A", I260 < 0), I260 * MM_Bottom) + IF (AND ($E260="C", T259 < 0, AE259 > 0), MIN (-T259, AE259) * 0.9916) - IF (E260="E", K260 * MM_Bottom)

This is an accounting of all U.S. dollars input into the AFFEERCE system.

This is a running total; the initial value for the period is (**AT259**). (**AR260**) are the dollar proceeds from the auction.

In the next operation (**IFERROR (IF ($E260="B", 2.49% * Q110 / (12 * Periods_Per_Month) * MM_Top, 0) * 99% / G260**), property owners buy their rent at an MM_TOP discount to peg from the market maker with the risk-free press of a button, so it is assumed that all rents will be paid this way. The average rent is 2.49% of the property value in the previous month(**Q110**) divided by 360 (**12 * Periods_Per_Month**), as rent is distributed monthly. The IFERROR handles a bogus last month's property value for the first month when the previous month does not exist on the spreadsheet.

The rent is multiplied by (**99%/ G260**), where **G260** is the current percentage of the peg where the Elsie trades. Typically, this value is 99% (the ratio is 1), except during deep discounting when rents are higher.

Treblers, to some extent, purchase Elsies for treble escrow to take advantage of the arbitrage. The number of expected treblers using Elsies is set in the parameter:

Property owners allowing rent to fall and percent of those treblers who use Elsies		50%

However, this column can ignore the parameter and takes the difference of the treble escrow Elsies in the current period and segment versus the number of treble-escrow Elsies in the previous period and segment to compute the dollars spent on the number of treble-escrow Elsies added during the current period (**(AF260 - AF259) * MM_Top**). If this number is negative, which will only occur on recovery from deep discounting, the market maker will purchase the Elsies at MM_Bottom. This error is not material to the outcome of the simulation.

Rescue mode purchases (**L260**) rarely bring dollars into the system, as they are properties purchased for the Elsies minted, with the U.S. dollars supplied by the donors. However, the property value in dollars (At the peg, dollars equal Elsies) is recorded in the dollar accounts, which is included as a dollar input in segment B. Both can be included, or both can be excluded in the accounting. I choose to include.

In the next operation (**IF ($E260="A", (Q260 - Q255) * retail_demand)**), (**Q260 - Q255**) is the is the difference between today's and yesterday's property value in the commons trust. (**retail_demand**) is the percent of property value added to the retail economy. The expression is the increased size of the retail trade denominated in dollars.

The next operation brings dollars into the system for market maker capitalization (**IF (AV260 < U260, MIN (U260 - AV260, AV260 * 0.1), 0)**). If the market maker's dollar inventory (**AV260**) is less than the market maker's Elsie inventory (**U260**), then increase capitalization by the difference (**U260 - AV260**) but by no more than a tenth of the current dollar inventory.

Retail supply (+) and demand shocks (-) that go through the market maker occur in segment E. In a supply shock, the market maker pays dollars for Elsies at MM_Bottom. Because this operation is subtracted, it represents a loss of dollars, as supply shocks are positive. In a demand shock, the market maker sells Elsies at MM_Top. Because demand shocks are negative, subtracting a negative represents a dollar increase. This is the entire operation: (**– IF (E260="E", IF (I260 < 0, I260 * MM_Top, I260 * MM_Bottom))**).

Demand shocks in segment A go through the ABC during ram and jam rather than the market maker. Dollars coming into the system during ram and jam come in at

MM_Bottom. This then is the next operation (**–IF (AND (E260="A", I260 < 0), I260 * MM_Bottom)**).

When Elsies are purchased in the 99.16% inventory, dollars received go directly to the distributor for delayed distribution. In segment C (**IF (AND ($E260="C", T259 < 0, AE259 > 0), MIN (-T259, AE259) * 0.9916)**), dollars are taken from delayed distribution sales when market maker inventory is depleted (**T259 < 0**), and there are Elsies in the 99.16% inventory (**AE259 > 0**). The extent of the sale is the lesser of the market maker deficit or the size of the 99.16% inventory (**MIN (-T259, AE259)**) multiplied by the price paid $0.9916/Elsie.

In the final operation (**- IF (E260="E", K260 * MM_Bottom)**) dollars are added to purchase Elsies for destruction. Because Elsies purchased for destruction (**K260**) is a negative number, the operation is subtracted for a positive addition of dollars.

MM Net

=AV260 + T260 * MM_Bottom

MM Net shows the total value of the market maker inventories. It is the sum of the dollar inventory plus the Elsie inventory multiplied by MM_Bottom (99.05%). It should always increase. Most of the increase is new capitalization, not profit. In an implementation, that would not be the case.

In the 20-year Phase I spreadsheet, the inventory is initialized to $6 million. Additional capitalization is required if the dollar inventory would go negative. The driving force for controlling inventory in the simulation is the Elsie desired inventory column, whose formulas can be manually overridden without harming spreadsheet balance.

Period	Y	M	D	Segment	Market Maker (Mil LC$)	MM Desired Inventory	MM Net	MM Dollars (Mil $)	Market Maker New Capital (Mil $)
0		0	0	E	0.00	6.00		$6.00	$0.00
1	2025	1	1	A	1.64	6.00	**$6.00**	$4.38	$0.44
1	2025	1	1	B	1.64	6.00	**$6.44**	$4.82	$0.48
1	2025	1	1	C	1.64	6.00	**$6.92**	$5.31	$0.53
1	2025	1	1	D	1.64	6.00	**$7.46**	$5.84	$0.16
1	2025	1	1	E	1.64	6.00	**$7.62**	$6.00	$0.00
2	2025	1	2	A	2.76	6.00	**$7.62**	$4.89	$0.49
2	2025	1	2	B	2.76	6.00	**$8.11**	$5.38	$0.54
2	2025	1	2	C	2.76	6.00	**$8.65**	$5.92	$0.08
2	2025	1	2	D	2.76	6.00	**$8.73**	$6.00	$0.00
2	2025	1	2	E	2.76	6.00	**$8.73**	$6.00	$0.00
3	2025	1	3	A	2.76	6.00	**$8.73**	$6.00	$0.00
3	2025	1	3	B	2.76	6.00	**$8.73**	$6.00	$0.00
3	2025	1	3	C	2.76	6.00	**$8.73**	$6.00	$0.00
3	2025	1	3	D	2.76	6.00	**$8.73**	$6.00	$0.00

The market maker began with $6 million to start. The desired inventory to start is 6 million Elsies

MM Net is the sum of the red-headed market maker Elsie column (x .9905) and the purple-headed MM Dollars column. It will always increase. The purple heading on the dollars column indicates that it is a repository of dollars whose sum must equal the total dollars inputted into the system.

Market Maker Dollar Account

=AV259 + IF ($E260="B", 57% * 95% * AR260, 0) – IF ($E260="D", V260 * MM_Bottom) -IF (AND ($E260="A", N260 > 0), (N260 - S260) * MM_Bottom, 0) + IF (E260="D", W260, 0) + IFERROR (IF ($E260="B", (2.49% * Q110 / (12 * Periods_Per_Month)) * MM_Top, 0) * 99% / G260 + (AF260 - AF259) * MM_Top, 0) + IF ($E260="A", (Q260 - Q255) * retail_demand) + AW259 – IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / MM_Bottom)), AH259 * MM_Bottom)) – IF (E260="E", IF (I260 < 0, I260 * MM_Top, I260 * MM_Bottom)) – IF (E260="E", K260 * MM_Bottom)

The dollar market-making account is the equal and opposite counterpart to the Elsie market-making account. When Elsies flow into the Elsie account, dollars flow out of this account, and vice versa.

The operations in one account match those in the other, except one and only one member of the pair has an MM_Top or MM_Bottom in the numerator or denominator.

The operations are shown in a table with the corresponding Elsie account operation. KD stands for a known dollar amount, and KE for a known Elsie amount.

Market Maker Operations Table			
Elsie Inventory	**Dollar Inventory**	**Direction**	**Function**
–IF ($E260="B", 2.49% * Q110 / (12 * Periods_Per_Month), 0) * 99%/G260	**+ IF ($E260="B", (2.49% * Q110 / (12 * Periods_Per_Month)) * MM_Top, 0) * 99% / G260**	D→KE	Property owner pays rent
+ IF (AND ($E260="A", N260 > 0), (N260 - S260), 0)	**–IF (AND ($E260="A", N260 > 0), (N260 - S260) * MM_Bottom, 0)**	KE→D	ABC mints and sells Elsies
- IF ($E260="B", 57% * 95% * AR260 / MM_Top, 0)	**+ IF ($E260="B", 57% * 95% * AR260, 0)**	KD->E	Convert auction proceeds to Elsies
+ IF ($E260="D", V260, 0)	**– IF ($E260="D", V260 * MM_Bottom)**	KE→D	Convert Elsies in rent payments to dollars
– IF (E260="D", W260/MM_Top, 0)	**+ IF (E260="D", W260, 0)**	KD→E	Convert dollars in rent payments to Elsies
- (AF260 – AF259)	**+ (AF260 - AF259) * MM_Top, 0)**	D→KE	Treblers add to escrow
– IF ($E260="A", (Q260 - Q255) * retail_demand / MM_Top)	**+ IF ($E260="A", (Q260 - Q255) * retail_demand)**	KD →E	Increasing retail trade
+ IF (E260="E", I260)	**– IF (E260="E", IF (I260 < 0, I260 * MM_Top, I260 * MM_Bottom))**	KE → D, D → KE	Demand shock or supply shock in Elsies
+ IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / 99.05)), AH259))	**–IF ($E260="A", IF (AND (O259 < F259, OR (O259 < 4, O259 < (F259 * S259) / 99.05)), AH259 * MM_Bottom))**	KE → D	Elsie land fund to dollar land fund
+ IF (E260="E", K260)	**– IF (E260="E", K260 * MM_Bottom)**	KE → D	Purchase Elsies to destroy
+ IF ($E260="A", AI259 * T257 / (T257 + AD257))	**-**	-	Market maker dividend
-	**+ AW259**	-	Market maker capitalization

Like the Elsie market maker that receives dividends, the dollar market maker also has one non-market maker operation. It is capitalized from a formula in AW259. The ABC is expected to use revenue over operations costs to fund its market maker, but that is not included in the spreadsheet.

Market Maker New Capital

=IF (AV260 < U260, MAX (MIN (U260 - AV260, AV260 * 0.1), 0), 0) + IF (AV260 < 0, -AV260 + 1, 0)

In an implementation, if the Elsie inventory is turning over daily and many opportunities are being missed, market maker capitalization should be increased. In a spreadsheet, judgment calls on inventory depletion from one-time large events versus normal depletion are challenging to program. Under some circumstances, this formula will overcapitalize the market maker, falsely shortening Phase I. However, if overcapitalized, the market maker will not recapitalize for longer, so the duration error is insignificant. Overcapitalization has more of an effect on market maker margins.

If the market maker dollar inventory is less than the market maker Elsie inventory (**AV260 < U260**), increase capitalization by the difference (**U260 - AV260**) or 10% of the dollar inventory (**AV260 * 0.1**), whichever is less. But not if (**U260 - AV260**) is less than zero.

If the market maker dollar inventory is less than 0 (**IF (AV260 < 0, -AV260 + 1, 0)**), increase the stock by the deficit plus $1 million.

Contract Escrow Plus Purchases

=AX259 + N260 + L260 + IF (AND (E260="A", I260 < 0), -I260)

This purple-headed column shows the value of signed contracts and purchases. It is the only expenditure of U.S. dollars accounted for in the spreadsheet. Dollars that come into the system are paid out for property. All other expenses come from the operating budgets of the VTLM and ABC.

Added to the total purchases to date (**AX259**) are ram and jam contracts (**N260**), rescued properties (**L260**), and properties purchased through ram and jam from Elsie demand that bypasses the market maker (**IF (AND (E260="A", I260 < 0), -I260)**).

Period	Y	M	D	Segment	Contract Escrow+Purchases (Mil $)	ABC (Mil $)	VTLM (Mil $)	Counties (Mil $)	Land Account + Land Fund Loan(Mil $)	Total Account Dollars (Mil $)	ABC Revenue per period
0		0	0	E					$1.00	$7.00	
1	2025	1	1	A	$6	$0.00	$0.00	$0.00	$0.70	$12	0.00
1	2025	1	1	B	$6	$0.00	$0.00	$0.00	$0.70	$12	0.00
1	2025	1	1	C	$6	$0.00	$0.00	$0.00	$0.70	$13	0.00
1	2025	1	1	D	$6	$0.00	$0.00	$0.00	$0.70	$13	0.00
1	2025	1	1	E	$6	$0.00	$0.00	$0.00	$0.70	$13	0.00
2	2025	1	2	A	$10	$0.00	$0.00	$0.00	$0.41	$16	0.00
2	2025	1	2	B	$10	$0.00	$0.00	$0.00	$0.41	$17	0.00
2	2025	1	2	C	$10	$0.00	$0.00	$0.00	$0.41	$17	0.00
2	2025	1	2	D	$10	$0.00	$0.00	$0.00	$0.41	$17	0.00
2	2025	1	2	E	$10	$0.00	$0.00	$0.00	$0.41	$17	0.00
352	2025	12	22	A	$2,502	$55.05	$64.23	$275.25	$5.21	$2,970	0.00
352	2025	12	22	B	$2,502	$55.05	$64.23	$275.25	$5.21	$2,977	0.00
352	2025	12	22	C	$2,502	$55.05	$64.23	$275.25	$5.21	$2,977	0.00
352	2025	12	22	D	$2,502	$55.05	$64.23	$275.25	$5.21	$2,977	0.00
352	2025	12	22	E	$2,502	$55.47	$64.71	$277.35	$5.21	$2,978	0.42
353	2025	12	23	A	$2,516	$55.47	$64.71	$277.35	$4.83	$2,989	0.00
353	2025	12	23	B	$2,516	$55.47	$64.71	$277.35	$4.83	$2,997	0.00
353	2025	12	23	C	$2,516	$55.47	$64.71	$277.35	$4.83	$2,997	0.00
353	2025	12	23	D	$2,516	$55.47	$64.71	$277.35	$4.83	$2,997	0.00
353	2025	12	23	E	$2,516	$55.97	$65.29	$279.83	$4.83	$2,998	0.50

The table shows Contract Escrow + Purchases and several other purple-headed columns whose values are summed to show Total Account Dollars in green. Not all purple-headed columns, including advance rent dollar fund and market maker dollar inventory, are shown. Most dollars that enter the system end up as property purchases.

ABC

=AY259 + IF ($E260="E", Z259 * 6% / 43%, 0)

Using the projected Phase I conditions in the above table (under Contract Escrow + Purchases), the ABC has a total accumulated revenue of $64.23 million by the 22nd day of month 12. As a purple-headed column, it must accumulate revenue. The final column breaks out the daily income of the ABC and VTLM, which receive the same distribution. The ABC receives daily revenue of $420,000.

The ABC brings in 6% of the ground rent (including purchasing agent distributions) or 6% of the revenue in the dollar distributor(**Z259**). This is 43% (6% ABC, 7% VTLM, 30% county) and is paid in segment E. The distributor distributes ground rent from the advance rent funds, 95% of dollar auction proceeds, and sales from the 99.16% delayed disbursement inventory.

The projected ABC revenue in 20 years is over $1 trillion. In the Phase I simulation, almost the same amount is in the 99.16% delayed disbursement inventory as

Phase I ends. Once Phase II begins, that inventory will quickly become depleted. In an implementation, logistical and political obstacles frequently empty the delayed inventory.

VTLM

```
=AZ259 + IF ($E260="E", Z259 * 7% / 43%, 0)
```

The VTLM (VIP Treasury and Land Management) is a not-for-profit organization in charge of the Commons Trust charters, land management, and the currency. All its funding goes for operations, including installing networks, if needed, and biometric readers (free, in some cases).

A conflict of interest will likely develop if started by the same collective as the ABC. This should be avoided. Government regulation of the VTLM is sought.

The VTLM is paid in U.S. dollars during segment E, with 7% out of the percent of funds in the dollar distributor (**Z259**), which holds 43% of all rent distributions. This is ground rent from the advance rent funds, 95% of dollar auction proceeds, and sales from the 99.16% delayed disbursement inventory.

Counties

```
=BA259 + IF ($E260="E", Z259 * 30% / 43%, 0)
```

The spreadsheet does not distinguish between counties. In an implementation, each county receives rent revenue only from properties in that county and auction proceeds only from auctions of properties in that county. The same holds for any jurisdiction where the agreement is signed.

Counties receive 30% of the ground rent or 30% out of the percent of funds in the dollar distributor(**Z259**).

For auction bids in cash on 40% land share property, the county receives 30% of 95% of 50%, or over 14% of the purchase price at closing (plus other closing costs that go to the county). They then receive 30% of 2.49% of the purchase price annually, or 0.747%. For Colorado and California counties, this is a windfall. However, even counties with as much as a 2% property tax will be ahead for over

ten years. During that time, the developmental growth of properties with no property tax will more than compensate for the lost revenue.

Land Account + Loan Account

=O260 + P260

This account is just what it claims to be. It is the sum of the land account (**O260**) and the deficit loan account (**P260**). Unlike those accounts, it is a purple-headed column that computes total dollar accounts.

The land fund is both a U.S. dollar account and an Elsie account. It is the U.S. dollar account that is summed.

Total Dollar Accounts

=SUM (AV260:BB260) + SUM (X260:Z260)

This is the sum of all dollar accounts in the AFFEERCE business plan, including all property purchased. To be balanced, the value here must always equal the value in total dollars input.

AV260	Market maker dollar account
AW260	Market maker new capital
AX260	Contracts and properties purchased
AY260	ABC operations and profit
AZ260	VTLM operations
BA260	County revenue
BB260	Land fund and loan account
X260	Advance rent fund (dollars)
Y260	Dollar rent
Z260	Dollar distributor

ABC Daily Revenue

```
=IF ($E261="E", Z260 * 6% / 43%, 0)
```

Same as the ABC purple-headed column, but not accumulating. Records the 6% ground rent distributions for the ABC and purchasing agents.